<?xml version="1.0" encoding="UTF-8"?>

<!--TREEBOLIC 2 INDENTED TXT- 2008/07/08  (C) 2002   Author: Bernard Bou-->

<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
  <xsl:output method="text" indent="no" encoding="UTF-8" omit-xml-declaration="yes"/>

  <xsl:template match="/">
    <xsl:apply-templates select="./jFamilyTree"/>
  </xsl:template>

  <xsl:template match="jFamilyTree">
    <xsl:text>#label:id:backcolor:forecolor:img:link:comment&#xA;</xsl:text> 
    <xsl:apply-templates select="./tree"/>
  </xsl:template>  

  <xsl:template match="tree">
    <xsl:apply-templates select="./nodes"/>
  </xsl:template>  

  <xsl:template match="nodes">
    <xsl:apply-templates select="./node"/>
  </xsl:template>

  <xsl:template match="node">
    <xsl:for-each select="ancestor::node">  
		<xsl:text>  </xsl:text>  
	</xsl:for-each>
	<xsl:apply-templates select="./label/text()"/>
	<xsl:text>:</xsl:text>  
    <xsl:apply-templates select="./@id"/>
	<xsl:text>:</xsl:text>  
    <xsl:apply-templates select="./@backcolor"/>
	<xsl:text>:</xsl:text>  
    <xsl:apply-templates select="./@forecolor"/>
	<xsl:text>:</xsl:text>  
    <xsl:apply-templates select="./img/@src"/>
	<xsl:text>:</xsl:text>  
    <xsl:apply-templates select="./a/@href"/>
	<xsl:text>:</xsl:text>  
    <xsl:apply-templates select="./content/text()"/>
    <xsl:text>&#xA;</xsl:text>
    <!-- recurse -->
    <xsl:apply-templates select="./node"/>
  </xsl:template>

  <xsl:template match="text()">
    <xsl:value-of select="normalize-space(.)"/>
  </xsl:template>

</xsl:stylesheet>
